```
In[1]:= << HolonomicFunctions.m
```

HolonomicFunctions package by Christoph Koutschan, RISC-Linz, Version 1.6 (12.04.2012)
⟶ Type ?HolonomicFunctions for help

Define the coefficient sequence of the series under consideration

```
In[75]:= ClearAll[F1, F2, F3, F4];
F1[n_Integer] := If[n == 0, 1 / x / y, F1[n] =
    Expand[((x + y + 1 / x + 1 / y) # - Coefficient[#, x, -1] - Coefficient[#, y, -1]) &[
      F1[n - 1]]]];
(*checked*) F2[n_Integer] := If[n == 0, 0, F2[n] =
    Expand[((x + y + 1 / x + 1 / y) # + Coefficient[F1[n - 1], y, -1] -
        Coefficient[#, x, -1] - 1 / y Coefficient[#, y, 0]) &[F2[n - 1]]]];
(*checked*) F3[n_Integer] := F2[n] /. {x → y, y → x};
F4[n_Integer] := If[n == 0, 0, F4[n] = Expand[
     ((x + y + 1 / x + 1 / y) # + Coefficient[F2[n - 1], x, -1] + Coefficient[F3[n - 1], y, -1] -
        1 / x Coefficient[#, x, 0] - 1 / y Coefficient[#, y, 0]) &[F4[n - 1]]]];
(*checked*)
```

```
In[5]:= Table[F1[n] + F2[n] + F3[n] + F4[n], {n, 0, 5}]
```

Out[5]=

$$\Big\{\frac{1}{x\,y},\ \frac{1}{x}+\frac{1}{x\,y^2}+\frac{1}{y}+\frac{1}{x^2\,y},\ 2+\frac{2}{x^2}+\frac{1}{x\,y^3}+\frac{2}{y^2}+\frac{2}{x^2\,y^2}+\frac{1}{x^3\,y}+\frac{2}{x\,y}+\frac{x}{y}+\frac{y}{x},$$

$$\frac{3}{x^3}+\frac{5}{x}+3\,x+\frac{1}{x\,y^4}+\frac{3}{y^3}+\frac{3}{x^2\,y^3}+\frac{3}{x^3\,y^2}+\frac{5}{x\,y^2}+\frac{3\,x}{y^2}+\frac{5}{y}+\frac{1}{x^4\,y}+\frac{5}{x^2\,y}+\frac{x^2}{y}+3\,y+\frac{3\,y}{x^2}+\frac{y^2}{x},$$

$$16+\frac{4}{x^4}+\frac{16}{x^2}+4\,x^2+\frac{1}{x\,y^5}+\frac{4}{y^4}+\frac{4}{x^2\,y^4}+\frac{6}{x^3\,y^3}+\frac{9}{x\,y^3}+\frac{6\,x}{y^3}+\frac{16}{y^2}+\frac{4}{x^4\,y^2}+$$

$$\frac{16}{x^2\,y^2}+\frac{4\,x^2}{y^2}+\frac{1}{x^5\,y}+\frac{9}{x^3\,y}+\frac{10}{x\,y}+\frac{9\,x}{y}+\frac{x^3}{y}+\frac{6\,y}{x^3}+\frac{9\,y}{x}+6\,x\,y+4\,y^2+\frac{4\,y^2}{x^2}+\frac{y^3}{x},$$

$$\frac{5}{x^5}+\frac{35}{x^3}+\frac{35}{x}+35\,x+5\,x^3+\frac{1}{x\,y^6}+\frac{5}{y^5}+\frac{5}{x^2\,y^5}+\frac{10}{x^3\,y^4}+\frac{14}{x\,y^4}+\frac{10\,x}{y^4}+\frac{35}{y^3}+\frac{10}{x^4\,y^3}+$$

$$\frac{35}{x^2\,y^3}+\frac{10\,x^2}{y^3}+\frac{5}{x^5\,y^2}+\frac{35}{x^3\,y^2}+\frac{35}{x\,y^2}+\frac{35\,x}{y^2}+\frac{5\,x^3}{y^2}+\frac{35}{y}+\frac{1}{x^6\,y}+\frac{14}{x^4\,y}+\frac{35}{x^2\,y}+$$

$$\frac{14\,x^2}{y}+\frac{x^4}{y}+35\,y+\frac{10\,y}{x^4}+\frac{35\,y}{x^2}+10\,x^2\,y+\frac{10\,y^2}{x^3}+\frac{14\,y^2}{x}+10\,x\,y^2+5\,y^3+\frac{5\,y^3}{x^2}+\frac{y^4}{x}\Big\}$$

```
In[33]:= PP[expr_, x_] := Expand[x * (Expand[1 / x expr] /. (x^i_ /; i < 0) → 0)];
(* positive part extraction tool *)
```

```
In[68]:= s = x + y + 1 / x + 1 / y; rat = (x y - x / y - y / x + 1 / x / y) / (1 - t s);
```

Derive a differential equation for F1(1,1,t)

In[70]:=
```
F1 == 1 / x / y HoldForm[PP][x, HoldForm[PP][y, rat]] ==
   1 / x / y RES[x, RES[y, 1 / x / y rat x / (1 - x) y / (1 - y)]] // TraditionalForm
```

Out[70]//TraditionalForm=

$$\mathrm{F1} = \frac{\mathrm{PP}\left(x, \mathrm{PP}\left(y, \frac{x\,y - \frac{x}{y} - \frac{y}{x} + \frac{1}{x\,y}}{1 - t\left(x + \frac{1}{x} + y + \frac{1}{y}\right)}\right)\right)}{x\,y} = \frac{\mathrm{RES}\left(x, \mathrm{RES}\left(y, \frac{x\,y - \frac{x}{y} - \frac{y}{x} + \frac{1}{x\,y}}{(1-x)\,(1-y)\left(1 - t\left(x + \frac{1}{x} + y + \frac{1}{y}\right)\right)}\right)\right)}{x\,y}$$

In[71]:=
```
deqF1 = First[CreativeTelescoping[
    First[CreativeTelescoping[rat / (1 - x) / (1 - y), Der[x], {Der[y], Der[t]}]],
    Der[y], {Der[t]}]]
```

Out[71]= $\left\{ \left(-t^2 + 16\,t^4\right) D_t^3 + \left(-6\,t + 8\,t^2 + 128\,t^3\right) D_t^2 + \left(-6 + 28\,t + 224\,t^2\right) D_t + \left(12 + 64\,t\right) \right\}$

Derive a differential equation for F2(1,1,t)=F3(1,1,t)

In[40]:=
```
F2 == t / y NP[x, RES[y, (x y - x / y - y / x + 1 / x / y) / (1 - t (x + y + 1 / x + 1 / y))]
     HoldForm[PP][y, (y - 1 / y) / (1 - t (x + y + 1 / x + 1 / y))]] // TraditionalForm
```

Out[40]//TraditionalForm=

$$\mathrm{F2} = \frac{t\,\mathrm{NP}\left(x, \mathrm{PP}\left(y, \frac{y - \frac{1}{y}}{1 - t\left(x + \frac{1}{x} + y + \frac{1}{y}\right)}\right)\mathrm{RES}\left(y, \frac{x\,y - \frac{x}{y} - \frac{y}{x} + \frac{1}{x\,y}}{1 - t\left(x + \frac{1}{x} + y + \frac{1}{y}\right)}\right)\right)}{y}$$

In[23]:=
```
deqA1 = First[CreativeTelescoping[rat, Der[y], {Der[t]}]];
(* differential equation for the residue *)
deqA2 = First[CreativeTelescoping[
     Y / y^2 / (1 - Y / y) (y - 1 / y) / (1 - t s), Der[y], {Der[t]}]] /. Y → y;
(* differential equation for the positive part *)
deqA = DFiniteTimes[deqA1, deqA2];
(* differential euation for the product of residue and positive part *)

aeqA = (-1 + x) ^2 * x * (1 + x) ^2 * y^2 + (-1 + x) * (1 + x) * y *
     (t^2 - 2 * t * x + x^2 - 2 * t * x^3 + t^2 * x^4 + t^2 * x * y - t * x^2 * y + t^2 * x^3 * y) * A +
    t^3 * x^2 * (t * x + t * y - x * y + t * x^2 * y + t * x * y^2) * A^2;
 (* guessed algebraic equation for the same quantity *)
Together[OreReduce[First[deqA],
  Annihilator[Root[Function[A, Evaluate[aeqA]], 1], {Der[t]}]]]
(* algebraic function satisfies differential equation
 derived above if this returns zero *)
```

Out[27]= 0

In[34]:=
```
aeqAsol = Sum[a[n] t^n, {n, 0, 10}] /. First[
    Solve[CoefficientList[aeqA /. A → (Sum[a[n] t^n, {n, 0, 10}] + O[t]^11), t] == 0,
     Table[a[n], {n, 0, 11}]]];
(* first terms of algebraic series solution of aeqA *)
truesol = Normal[Coefficient[Normal[Series[rat, {t, 0, 10}]], y, -1]
     PP[Normal[Series[(y - 1 / y) / (1 - t s), {t, 0, 10}]], y] + O[t]^11];
(* first terms of F2(1,1,t) *)
aeqAsol - truesol // Together (* if this returns zero,
 F2(1,1,t) agrees with the algebraic function defined by aeqA *)
```

Solve: Equations may not give solutions for all "solve" variables.

Out[36]= 0

In[37]:=
```
aux = Collect[
   Numerator[Together[Evaluate[aeqA /. A → A / (t / (1 - x)) /. y → 1]]], A, Factor];
deqF2 = First[CreativeTelescoping[Annihilator[
    Root[Function[A, Evaluate[aux]], 1], {Der[x], Der[t]}], Der[x], {Der[t]}]]
 (* differential equation for F2(1,1,t) *)
```

Out[38]= $\{(-t^3 + 16\,t^5)\,D_t^3 + (-7\,t^2 + 8\,t^3 + 144\,t^4)\,D_t^2 + (-6\,t + 36\,t^2 + 288\,t^3)\,D_t + (6 + 24\,t + 96\,t^2)\}$

In[39]:=
```
Together[ApplyOreOperator[First[deqF2],
   (Sum[F2[n] t^n, {n, 0, 10}] + O[t]^11) /. {x → 1, y → 1}]]
(* cross check, should give zero *)
```

Out[39]= $O[t]^{11}$

Derive a Differential Equation for F4

In[41]:=
```
F4 == 1 / x / y t^2 RES[y, 1 / y^2 / (1 - 1 / y) RES[x,  (y - 1 / y) / (1 - t s) RES[y,  rat]]
     HoldForm[PP][x, (x - 1 / x) / (1 -  t s)]] // TraditionalForm
```

Out[41]//TraditionalForm=

$$\mathrm{F4} = \frac{t^2\,\mathrm{RES}\left(y, \frac{\mathrm{PP}\left(x, \frac{x-\frac{1}{x}}{1-t\left(x+\frac{1}{x}+y+\frac{1}{y}\right)}\right)\mathrm{RES}\left(x, \frac{\left(y-\frac{1}{y}\right)\mathrm{RES}\left(y, \frac{x\,y-\frac{x}{y}-\frac{y}{x}+\frac{1}{x\,y}}{1-t\left(x+\frac{1}{x}+y+\frac{1}{y}\right)}\right)}{1-t\left(x+\frac{1}{x}+y+\frac{1}{y}\right)}\right)}{\left(1-\frac{1}{y}\right)y^2}\right)}{x\,y}$$

```
deqf = First[CreativeTelescoping[rat, Der[y], {Der[x], Der[t]}]];
(* differential equations for innermost residue, as a function of x and t *)
deqf = OreGroebnerBasis[ChangeOreAlgebra[#, OreAlgebra[Der[x], Der[t], Der[y]]] & /@
    Append[deqf, ToOrePolynomial[Der[y]]]];
(* view the innermost residue also as a (constant)
 function with respect to y *)
deqf = First[CreativeTelescoping[
   DFiniteTimesHyper[deqf, (y - 1 / y) / (1 - t s)], Der[x], {Der[y], Der[t]}]];
(* differential equations for the middle residue;
this takes some time *)
```

The system of differential equations derived above is not minimal. Via guessing, we can find that the middle residue satisfies some additional differential equations that are not consequences of the derived equations (i.e.: not elements of the ideal generated by the corresponding operators). Using closure properties and comparison of initial values, we can certify that these guessed operators are correct, so we are entitled to add them to the ideal. Doing so will speed up some of the subsequent computations.

```
In[56]:= deqfguess = ToOrePolynomial[#, OreAlgebra[Der[y], Der[t]]] & /@
    (* additional guessed operators *){-2 * t - 3 * y + 96 * t^2 * y + 18 * t * y^2 -
      18 * y^3 - 192 * t^2 * y^3 + 18 * t * y^4 - 3 * y^5 + 96 * t^2 * y^5 - 2 * t * y^6 +
      t * (-1 + y) ^2 * (1 + y) ^2 * (-2 * t - 5 * y + 96 * t^2 * y - 2 * t * y^2) * Der[t] +
      t^2 * (-1 + 4 * t) * (1 + 4 * t) * (-1 + y) ^2 * y * (1 + y) ^2 * Der[t] ^2 - (-1 + y) * (1 + y) *
       (2 * t - 3 * y + 2 * t * y^2) * (4 * t + 3 * y - 8 * t * y^2 + y^3 + 4 * t * y^4) * Der[y] -
      2 * t^2 * (-1 + y) ^3 * (1 + y) ^3 * (2 * t - 3 * y + 2 * t * y^2) * Der[t] * Der[y] +
      (-1 + y) ^2 * y^2 * (1 + y) ^2 * (2 * t - 3 * y + 2 * t * y^2) * Der[y] ^2,
     t^2 - 12 * t * y + 16 * t^3 * y + 109 * t^2 * y^2 - 48 * t * y^3 - 64 * t^3 * y^3 +
      24 * y^4 - 110 * t^2 * y^4 - 72 * t * y^5 + 96 * t^3 * y^5 + 24 * y^6 - 110 * t^2 * y^6 -
      48 * t * y^7 - 64 * t^3 * y^7 + 109 * t^2 * y^8 - 12 * t * y^9 + 16 * t^3 * y^9 + t^2 * y^10 +
      (t^3 - 4 * t^2 * y + 8 * t^4 * y + 37 * t^3 * y^2 - 4 * t^2 * y^3 - 32 * t^4 * y^3 -
         38 * t^3 * y^4 + 16 * t^2 * y^5 + 48 * t^4 * y^5 - 38 * t^3 * y^6 - 4 * t^2 * y^7 -
         32 * t^4 * y^7 + 37 * t^3 * y^8 - 4 * t^2 * y^9 + 8 * t^4 * y^9 + t^3 * y^10) * Der[t] +
      (-4 * t^3 + 19 * t^2 * y - 24 * t * y^2 + 16 * t^3 * y^2 + 6 * y^3 + 23 * t^2 * y^3 -
         12 * t * y^4 - 20 * t^3 * y^4 + 12 * y^5 - 170 * t^2 * y^5 + 12 * t * y^6 -
         18 * y^7 + 182 * t^2 * y^7 + 12 * t * y^8 + 20 * t^3 * y^8 - 41 * t^2 * y^9 +
         12 * t * y^10 - 16 * t^3 * y^10 - 13 * t^2 * y^11 + 4 * t^3 * y^12) * Der[y] +
      (-2 * t^4 + 7 * t^3 * y - 5 * t^2 * y^2 + 8 * t^4 * y^2 + 5 * t^3 * y^3 +
         10 * t^2 * y^4 - 10 * t^4 * y^4 - 50 * t^3 * y^5 + 50 * t^3 * y^7 -
         10 * t^2 * y^8 + 10 * t^4 * y^8 - 5 * t^3 * y^9 + 5 * t^2 * y^10 -
         8 * t^4 * y^10 - 7 * t^3 * y^11 + 2 * t^4 * y^12) * Der[t] * Der[y] +
      (6 * t^2 * y^2 - 12 * t * y^3 + 6 * y^4 - 24 * t^2 * y^4 + 12 * t * y^5 - 12 * y^6 +
         36 * t^2 * y^6 + 12 * t * y^7 + 6 * y^8 - 24 * t^2 * y^8 - 12 * t * y^9 + 6 * t^2 * y^10) *
       Der[y] ^2 + (t^2 * y^3 - 2 * t * y^4 + y^5 - 5 * t^2 * y^5 + 4 * t * y^6 -
         3 * y^7 + 10 * t^2 * y^7 + 3 * y^9 - 10 * t^2 * y^9 - 4 * t * y^10 -
         y^11 + 5 * t^2 * y^11 + 2 * t * y^12 - t^2 * y^13) * Der[y] ^3};
     UnderTheStaircase[DFinitePlus @@ (DFiniteOreAction[deqf, #] & /@ deqfguess)]
     (* initial values that have to be checked in order to confirm the correctness
      of the guessed operators. Since these terms were used for the guessing,
     the initial values match by construction so we do not perform a check here *)

Out[57]= {1, D_t, D_y}

In[58]:= deqf = OreGroebnerBasis[Join[deqf, deqfguess]];
     (* include the guessed equations into the system
      of differential equations for the middle residue *)

In[59]:= deqg = First[CreativeTelescoping[Annihilator[1 / x^2 / (1 - 1 / x) (x - 1 / x) / (1 - t s),
         {Der[y], Der[x], Der[t]}], Der[x], {Der[y], Der[t]}]];
     (* differential equations for the positive part *)
```

```
deqF4 = First[FindCreativeTelescoping[DFiniteTimesHyper[
    DFiniteTimes[deqf, deqg], t^2 / y^2 / (1 - 1 / y)], {Der[y]}, {Der[t]}]]
  (* differential equation for the outer residue; this takes some time *)
```

Out[64]=

$$\{(-225 t^5 - 480 t^6 + 12\,400 t^7 + 7680 t^8 - 349\,440 t^9 + 344\,064 t^{10} + 6\,893\,568 t^{11} - 8\,650\,752 t^{12} - 80\,936\,960 t^{13} + 44\,040\,192 t^{14} + 374\,341\,632 t^{15} + 100\,663\,296 t^{16} + 150\,994\,944 t^{17} + 268\,435\,456 t^{19}) D_t^8 +$$
$$(-8100 t^4 - 11\,400 t^5 + 541\,120 t^6 + 272\,640 t^7 - 15\,406\,080 t^8 + 15\,673\,344 t^9 + 298\,696\,704 t^{10} - 501\,350\,400 t^{11} - 3\,897\,294\,848 t^{12} + 3\,276\,275\,712 t^{13} + 21\,428\,699\,136 t^{14} + 5\,788\,139\,520 t^{15} + 8\,657\,043\,456 t^{16} + 402\,653\,184 t^{17} + 13\,958\,643\,712 t^{18}) D_t^7 +$$
$$(-100\,800 t^3 - 31\,860 t^4 + 8\,557\,760 t^5 + 2\,094\,720 t^6 - 258\,954\,240 t^7 + 250\,512\,384 t^8 + 4\,804\,902\,912 t^9 - 10\,314\,645\,504 t^{10} - 68\,717\,903\,872 t^{11} + 86\,525\,607\,936 t^{12} + 462\,359\,101\,440 t^{13} + 125\,199\,974\,400 t^{14} + 184\,616\,484\,864 t^{15} + 15\,502\,147\,584 t^{16} + 272\,730\,423\,296 t^{17}) D_t^6 +$$
$$(-529\,200 t^2 + 799\,560 t^3 + 61\,569\,600 t^4 - 16\,231\,680 t^5 - 2\,131\,246\,080 t^6 + 1\,797\,765\,120 t^7 + 37\,058\,936\,832 t^8 - 95\,536\,152\,576 t^9 - 557\,805\,207\,552 t^{10} + 1\,057\,035\,386\,880 t^{11} + 4\,793\,674\,235\,904 t^{12} + 1\,296\,593\,584\,128 t^{13} + 1\,868\,109\,447\,168 t^{14} + 213\,808\,840\,704 t^{15} + 2\,551\,210\,573\,824 t^{16}) D_t^5 +$$
$$(-1\,134\,000 t + 5\,886\,000 t^2 + 207\,596\,400 t^3 - 260\,784\,000 t^4 - 9\,107\,654\,400 t^5 + 6\,493\,132\,800 t^6 + 149\,710\,233\,600 t^7 - 412\,183\,756\,800 t^8 - 2\,146\,284\,994\,560 t^9 + 6\,395\,768\,340\,480 t^{10} + 25\,267\,446\,743\,040 t^{11} + 6\,790\,620\,119\,040 t^{12} + 9\,489\,277\,255\,680 t^{13} + 1\,322\,715\,709\,440 t^{14} + 12\,079\,595\,520\,000 t^{15}) D_t^4 +$$
$$(-756\,000 + 12\,348\,000 t + 303\,852\,000 t^2 - 1\,023\,984\,000 t^3 - 19\,521\,446\,400 t^4 + 14\,295\,859\,200 t^5 + 328\,041\,676\,800 t^6 - 758\,086\,041\,600 t^7 - 3\,579\,000\,913\,920 t^8 + 18\,840\,299\,765\,760 t^9 + 66\,076\,048\,097\,280 t^{10} + 17\,517\,300\,940\,800 t^{11} + 23\,589\,940\,101\,120 t^{12} + 3\,768\,833\,802\,240 t^{13} + 28\,217\,935\,134\,720 t^{14}) D_t^3 +$$
$$(6\,804\,000 + 154\,224\,000 t - 1\,396\,872\,000 t^2 - 18\,370\,368\,000 t^3 + 22\,303\,641\,600 t^4 + 377\,981\,337\,600 t^5 - 416\,201\,932\,800 t^6 - 1\,842\,079\,334\,400 t^7 + 24\,572\,020\,654\,080 t^8 + 77\,591\,771\,873\,280 t^9 + 20\,092\,708\,454\,400 t^{10} + 25\,915\,262\,238\,720 t^{11} + 4\,566\,087\,106\,560 t^{12} + 29\,377\,576\,304\,640 t^{13}) D_t^2 +$$
$$(13\,608\,000 - 544\,320\,000 t - 5\,553\,792\,000 t^2 + 16\,809\,984\,000 t^3 + 186\,745\,651\,200 t^4 + 70\,778\,880\,000 t^5 + 221\,655\,859\,200 t^6 + 11\,344\,438\,886\,400 t^7 + 32\,603\,017\,052\,160 t^8 + 8\,131\,077\,734\,400 t^9 + 9\,974\,726\,000\,640 t^{10} + 1\,884\,416\,901\,120 t^{11} + 10\,823\,317\,585\,920 t^{12}) D_t +$$
$$(-27\,216\,000 - 217\,728\,000 t + 2\,626\,560\,000 t^2 + 21\,012\,480\,000 t^3 + 31\,275\,417\,600 t^4 + 103\,926\,988\,800 t^5 + 1\,026\,293\,760\,000 t^6 + 2\,706\,584\,371\,200 t^7 + 635\,311\,226\,880 t^8 + 733\,835\,427\,840 t^9 + 144\,955\,146\,240 t^{10} + 773\,094\,113\,280 t^{11})\}$$

In[61]:=

```
ApplyOreOperator[First[deqF4], Sum[F4[n] t^n, {n, 0, 15}] + O[t]^15] /.
  {x → 1, y → 1} (* cross check,should give zero *)
```

Out[61]= $O[t]^{12}$

Combine the derived differential operators to one for F.

In[89]:=

```
deqF = DFinitePlus[deqF1, deqF2, deqF4]; (* certified operator; it has order 8 *)
```

```
deqFguessed =
  ToOrePolynomial[60 + 408 t + 672 t^2 + 384 t^3 + (-15 + 6 t + 600 t^2 + 1584 t^3 + 1152 t^4) Der[t] +
    (-9 t - 24 t^2 + 180 t^3 + 672 t^4 + 576 t^5) Der[t]^2 + (-t^2 - 4 t^3 + 12 t^4 + 64 t^5 + 64 t^6) Der[t]^3];
(* guessed shorter operator *)

In[88]:= OreReduce[deqF, {deqFguessed}] (* if this gives zero,
the guessed operator is correct *)

Out[88]= {0}

In[110]:= recF = DFiniteDE2RE[{deqFguessed}, {t}, {n}];
(* certified recurrence operator for the coefficients of F(1,1,t);
it has order 4 *)

In[111]:= recFguessed = ToOrePolynomial[-192 - 736*n - 992*n^2 - 592*n^3 - 160*n^4 - 16*n^5 +

In[112]:= OreReduce[recF, {recFguessed}] (* if this is zero,
the guessed operator is correct *)

Out[112]= {0}

In[113]:= << Asymptotics.m

Asymptotics  Package by Manuel Kauers — © RISC Linz — V 0.11 (2012-07-19)

Asymptotics[ApplyOreOperator[recFguessed, f[n]], f[n]]
(* negative integer exponents of n prove transcendence of generating functions *)
```

Out[114]= $\left\{\frac{(-4)^n}{n^3}, \frac{4^n}{n}\right\}$